

Specifications:

The goal of the project was to design an inexpensive filter that provides -42dB/octave of attenuation with -3dB cutoff frequency at 40MHz. The filter response in the pass-band must be maximally flat with little ripple in the stop-band. GALFA spectrometer application requires 14 filter pairs where each pair of filters is phase matched to within three degrees.

Design and simulation:

LC ladder filter design is a well covered subject, and books such **Handbook of Filter Synthesis**, John Wiley, 1967 by A. I. Zverev cover the topic thoroughly. To meet the specification, 7th order Butterworth filter was chosen because it is flat in the pass-band and has no ripple in the stop-band. To calculate the L and C value, the web applet at http://www.wa4dsy.net/filter/hp_lp_filter.html was used. The applet requires the user to input the filter order, -3dB cutoff frequency, and input/output impedance. In this case, the filter is 7th order, with 40MHz -3dB cutoff frequency with 49 Ω input/output impedance (49 Ω I/O impedance was chosen because it gives closest commercially available L and C values). Fig.1 shows the applet output.




Figure 1

To insure that these results are correct, HSPICE was used to check (when doing HSPICE modeling do not forget to put a 49 Ω resistor in series with voltage source driving the input, and 49 Ω resistor shunting the output). Fig. 2 shows the magnitude and phase of the filter for the part values given by the applet. The measurement shows that the -3dB cutoff frequency is exactly at 40MHz. Notice that the low frequency output is at 0 dB, however, due to the resistive divider it should be at -6dB (the output was multiply by 2 to correspond to network analyzer measurements). The frequency scale is logarithmic, and the phase wraps around.




Figure 2

It must be kept in mind that L and C values have some tolerances and that two filters in a pair will never exactly match. Therefore it is crucial to use a Monte Carlo simulation to understand the possible phase spread among filters at the frequency of importance. Fig.3 shows a Monte Carlo simulation with 30 trials where L's and C's have 1% tolerance (Gaussian distribution). The HSPICE simulation shows that at 33MHz the worst case phase spread is approximately two degrees.




Figure 3

Parts, schematic, and board:

This design used 0805 (size) ceramic chip capacitors from Kemet Electronics with 1% tolerance. The capacitors were made out of C0G (NP0) material. The inductors were also 0805's with 2% tolerances from Coilcraft Inc. Table 1 shows the list of parts used (note they are not exactly our desired values but the closest commercially available values). C2/C3 is comprised of two capacitors in parallel. Note that L1=L2, C1=C4, C2=C3. 0805 package was chosen due to part availability and ease of soldering.

Part	Value	Part Reference
L1	240nH	0805CS-241XGBB
L2	390nH	0805CS-391XGBB
L2	240nH	0805CS-241XGBB
C1	36pF	C0805C360F8GAC3810
C2	100pF+47pF	C0805C101F8GAC3810 + C0805C470F8GAC3810
C3	100pF+47pF	C0805C101F8GAC3810 + C0805C470F8GAC3810
C3	36pF	C0805C360F8GAC3810

Table 1

Fig. 4 shows Orcad Capture schematic of the filter. Use of TEST POINT's will be explained in next paragraph.




Figure 4

It was desired that this filter board can be interchanged with a commercial filter from Synergy Microwave that used 124 style package (Fig 4). Therefore, TEST POINT's are used for as drill holes where headers can be soldered and then piggy backed to the main GALFA board.




Figure 5

Fig. 6 shows the Orcad Layout Plus of the filter board. The board size is 0.8" by 0.8". The top left drill hole is filter input, and the bottom right hole is the filter output (due to symmetry, the input/output can be flipped). The other holes are connected to the ground layer. All holes correspond to pins in 124 style package (200 mils apart). All capacitor pads that go to ground are connected by one or more vias. To prevent mutual inductance, Coilcraft engineer suggested that inductors should be layout out perpendicularly to one another, or as far apart as possible (at least 100 mils). L2 is perpendicular to L1 and L3, and L1 and L3 are fairly far part. The board traces are 20 mils wide and do not have 90 degree turns. The board was manufactured by Advanced Circuits. The price for 40 two layer boards with 1oz copper, no solder mask, and no silk screen was 200 dollars with 3 day turn around.




Figure 6

Test Results:

HP 8753C network analyzer was used to test two filter boards. Fig. 7 shows magnitude response (in dB) of filter 1 and fig. 8 shows magnitude response (in dB) of filter 2. The sweep was done from 300kHz to 100MHz.




Figure 7, Filter 1




Figure 8, Filter 2

The insertion loss of both filter is approximately -0.25 dB, and the -3dB cutoff frequency is 37.5MHz. The cutoff frequency is lower than what the filter was designed for because the parasitic capacitances of the pads on the board were not considered. The parasitic pad capacitances can be modeled as 1.5pF caps in parallel with each capacitor in schematic (C1, C2, C3, and C4). Because the parasitic capacitance is almost uniform from one board to the other, the magnitude and phase difference is not affected. Table 2 lists the magnitude, phase, and their difference at several frequencies of interest. (Measurement were done with calibration).

Frequency	Mag. Filter 1	Mag Filter 2	\Delta Mag	\Phi Filter 1	\Phi Filter 1	\Delta\Phi
300kHz	-0.72dB	-0.73dB	0.01dB	14.9°	14.9°	0°
37MHz	-3.82dB	-3.88dB	0.06dB	-34.2°	-34.4°	0.2°
38MHz	-4.43dB	-4.39dB	0.04dB	-47.3°	-47.3°	0°
40MHz	-5.9dB	-5.85dB	0.05dB	-73.1°	-73.5°	0.4°

Table 2

Summary

Overall, the results show that the two filters with 1% tolerance parts can be easily phase matched to with 3 degrees, however, due to parasitic capacitances the -3dB is at 37.5 MHz instead of 40MHz. The pass band filter response is flat, and there is indeed -42dB/octave of attenuation after the cutoff frequency.

*****The schematic and layout files can be found in /disks/glenn/a/users/pmonat directory.**

Parts datasheets:

Kemet capacitors:

[http://www.kemet.com/kemet/web/homepage/kechome.nsf/vapubfilename/F3102HCE.pdf/\\$file/F3102HCE.pdf](http://www.kemet.com/kemet/web/homepage/kechome.nsf/vapubfilename/F3102HCE.pdf/$file/F3102HCE.pdf)

Coilcraft inductors:

<http://www.coilcraft.com/0805cs.cfm>

HSPICE files:

Ideal 7th order Butterworth with 40 MHz cutoff with 49 Ω I/O:

Butterworth 7th order

```
*
.param RvarS=49          $unif(50, .01, 3)
.param RvarO=49          $unif(50, .01, 3)
.param Lvar1=0.243u       $unif(0.221u , .01, 3)
.param Lvar2=0.390u       $unif(0.354u , .01, 3)
.param Lvar3=0.243u       $unif(0.221u , .01, 3)
.param Cvar1=36.13p        $unif(31.48p , .01, 3)
.param Cvar2=146.32p       $unif(127.46p, .01, 3)
.param Cvar3=146.32p       $unif(127.46p, .01, 3)
.param Cvar4=36.13p        $unif(31.48p , .01, 3)
```

```
vi vi 0 ac 1
```

```
L1 1 2 Lvar1
L2 2 3 Lvar2
L3 3 vo Lvar3
C1 1 0 Cvar1
C2 2 0 Cvar2
C3 3 0 Cvar3
C4 vo 0 Cvar4
```

```

Rs vi 1 RvarS
Ro vo 0 RvarO

.option post=2 nomod
.ac dec 200 lx lg $sweep Lvar1 .240u '.243u' '3n'
$.print ac vdb(vo, vi) vp(vo,0)
.end

```

Monte Carlo simulation (30 runs) of 7th order Butterworth with 1% tolerance parts with Gaussian distribution:

```

Cheby 7th order Monte Carlo
*
.param RvarS=gauss(49, .01, 3)
.param RvarO=gauss(49, .01, 3)
.param Lvar1=gauss(0.243u , .01, 3)
.param Lvar2=gauss(0.390u , .01, 3)
.param Lvar3=gauss(0.243u , .01, 3)
.param Cvar1=gauss(36.13p , .01, 3)
.param Cvar2=gauss(146.32p, .01, 3)
.param Cvar3=gauss(146.32p, .01, 3)
.param Cvar4=gauss(36.13p , .01, 3)

```

```
vi vi 0 ac 1
```

```

L1 1 2 Lvar1
L2 2 3 Lvar2
L3 3 vo Lvar3
C1 1 0 Cvar1
C2 2 0 Cvar2
C3 3 0 Cvar3
C4 vo 0 Cvar4

```

```

Rs vi 1 RvarS
Ro vo 0 RvarO

.option post=2 nomod
.ac dec 500 lx lg sweep monte=30
$.print ac vdb(vo, vi) vp(vo,0)
.end

```